query: embed.html## map of Belgian items with article in exactly one of German-, French-, Dutch-language Wikipedias
  #defaultView:Map
  SELECT ?item (SAMPLE(?title) AS ?itemLabel) (SAMPLE(?location) AS ?location) (SAMPLE(?language) AS ?layer) WITH {
    SELECT * WHERE {
      wd:Q31 p:P1332/psv:P1332/wikibase:geoLatitude ?n;
             p:P1333/psv:P1333/wikibase:geoLatitude ?s;
             p:P1334/psv:P1334/wikibase:geoLongitude ?e;
             p:P1335/psv:P1335/wikibase:geoLongitude ?w.
    }
  } AS %belgiumBoundingBox WHERE {
    VALUES ?wiki { <https://de.wikipedia.org/> <https://fr.wikipedia.org/> <https://nl.wikipedia.org/> }
    ?item wdt:P17 wd:Q31;
          wdt:P625 ?location.
    ?article a schema:Article;
             schema:about ?item;
             schema:isPartOf ?wiki;
             schema:inLanguage ?language;
             schema:name ?title.
    # filter out some stray results that have country Belgium but coordinates outside it (e. g. rivers)
    INCLUDE %belgiumBoundingBox.
    BIND(geof:latitude(?location) AS ?lat)
    BIND(geof:longitude(?location) AS ?lon)
    FILTER(?s <= ?lat && ?lat <= ?n &&
           ?w <= ?lon && ?lon <= ?e)
  }
  GROUP BY ?item
  HAVING(COUNT(DISTINCT ?wiki) = 1)